


# Programming Languages and Compilers (CS 421)



Elsa L Gunter  
2112 SC, UIUC

<http://www.cs.uiuc.edu/class/cs421/>

Based in part on slides by Mattox Beckman, as updated by Vikram Adve and Gul Agha



# Recursion Example

---

Compute  $n^2$  recursively using:

$$n^2 = (2 * n - 1) + (n - 1)^2$$

```
# let rec nthsq n =      (* rec for recursion *)
  match n              (* pattern matching for cases *)
  with 0 -> 0          (* base case *)
  | n -> (2 * n - 1)   (* recursive case *)
      + nthsq (n - 1);; (* recursive call *)
val nthsq : int -> int = <fun>
# nthsq 3;;
- : int = 9
```

Structure of recursion similar to inductive proof



# Recursion and Induction

---

```
# let rec nthsq n = match n with 0 -> 0  
  | n -> (2 * n - 1) + nthsq (n - 1) ;;
```

- Base case is the last case; it stops the computation
- Recursive call must be to arguments that are somehow smaller - must progress to base case
- **if** or **match** must contain base case
- Failure of these may cause failure of termination



# Structural Recursion

---

- Functions on recursive datatypes (eg lists) tend to be recursive
- Recursion over recursive datatypes generally by structural recursion
  - Recursive calls made to components of structure of the same recursive type
  - Base cases of recursive types stop the recursion of the function



# Structural Recursion : List Example

---

```
# let rec length list = match list
  with [ ] -> 0 (* Nil case *)
       | x :: xs -> 1 + length xs;; (* Cons case *)
val length : 'a list -> int = <fun>
# length [5; 4; 3; 2];;
- : int = 4
```

- Nil case [ ] is base case
- Cons case recurses on component list xs



# Forward Recursion

---

- In structural recursion, you split your input into components
- Forward recursion limited form of structural recursion
- In forward recursion:
  - First call function recursively on all recursive components
  - Recursive calls use only components and input arguments, no other computation
  - Build result from results of component



# Forward Recursion: Examples

---

```
# let rec double_up list =  
  match list  
  with [ ] -> [ ]  
       | (x :: xs) -> (x :: x :: double_up xs);;  
val double_up : 'a list -> 'a list = <fun>
```

```
# let rec poor_rev list =  
  match list  
  with [] -> []  
       | (x::xs) -> poor_rev xs @ [x];;  
val poor_rev : 'a list -> 'a list = <fun>
```



# Mapping Recursion

---

- One common form of structural recursion applies a function to each element in the structure

```
# let rec doubleList list = match list  
  with [ ] -> [ ]  
       | x::xs -> 2 * x :: doubleList xs;;
```

```
val doubleList : int list -> int list = <fun>
```

```
# doubleList [2;3;4];;
```

```
- : int list = [4; 6; 8]
```





## Recall Map

---

```
# let rec map f list =
```

```
  match list
```

```
  with [] -> []
```

```
  | (h::t) -> (f h) :: (map f t);;
```

```
val map : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
# map plus_two fib5;;
```

```
- : int list = [10; 7; 5; 4; 3; 3]
```

■ Same as List.map



# Mapping Recursion

---

- Can use the higher-order recursive map function instead of direct recursion

```
# let doubleList list =
```

```
  List.map (fun x -> 2 * x) list;;
```

```
val doubleList : int list -> int list = <fun>
```

```
# doubleList [2;3;4];;
```

```
- : int list = [4; 6; 8]
```

- Same function, but no rec



# Folding Recursion

---

- Another common form “folds” an operation over the elements of the structure

```
# let rec multList list = match list  
  with [ ] -> 1  
       | x::xs -> x * multList xs;;
```

```
val multList : int list -> int = <fun>
```

```
# multList [2;4;6];;
```

```
- : int = 48
```

- Computes  $(2 * (4 * (6 * 1)))$



# Folding Recursion

---

- multList folds to the right
- Same as:

```
# let multList list =  
  List.fold_right  
    (fun p -> fun x -> x * p)  
    list 1;;
```

```
val multList : int list -> int = <fun>
```

```
# multList [2;4;6];;
```

```
- : int = 48
```



# Folding Functions over Lists

---

How are the following functions similar?

```
# let rec sumlist list = match list with  
  [ ] -> 0 | x::xs -> x + sumlist xs;;
```

```
val sumlist : int list -> int = <fun>
```

```
# sumlist [2;3;4];;
```

```
- : int = 9
```

```
# let rec prodlist list = match list with
```

```
  [ ] -> 1 | x::xs -> x * prodlist xs;;
```

```
val prodlist : int list -> int = <fun>
```

```
# prodlist [2;3;4];;
```

```
- : int = 24
```



# Folding

---

```
# let rec fold_left f a list = match list
  with [] -> a | (x :: xs) -> fold_left f (f a x) xs;;
val fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a =
  <fun>
```

```
fold_left f a [x1; x2;...;xn] = f(...(f (f a x1) x2)...)xn
```

```
# let rec fold_right f list b = match list
  with [] -> b | (x :: xs) -> f x (fold_right f xs b);;
val fold_right : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b =
  <fun>
```

```
fold_right f [x1; x2;...;xn] b = f x1(f x2 (...(f xn b)...))
```



# Folding - Forward Recursion

---

```
# let sumlist list = fold_right (+) list 0;;
```

```
val sumlist : int list -> int = <fun>
```

```
# sumlist [2;3;4];;
```

```
- : int = 9
```

```
# let prodlist list = fold_right ( * ) list 1;;
```

```
val prodlist : int list -> int = <fun>
```

```
# prodlist [2;3;4];;
```

```
- : int = 24
```

# Encoding Recursion with Fold

```
# let rec append list1 list2 = match list1 with  
  [ ] -> list2 | x::xs -> x :: append xs list2;;  
val append : 'a list -> 'a list -> 'a list = <fun>
```

Base Case

Operation

Recursive Call

```
# let append list1 list2 =  
  fold_right (fun x y -> x :: y) list1 list2;;  
val append : 'a list -> 'a list -> 'a list = <fun>  
# append [1;2;3] [4;5;6];;  
- : int list = [1; 2; 3; 4; 5; 6]
```





# Map from Fold


---

```
# let map f list =  
  fold_right (fun x y -> f x :: y) list [ ];;  
val map : ('a -> 'b) -> 'a list -> 'b list =  
  <fun>  
# map ((+)1) [1;2;3];;  
- : int list = [2; 3; 4]
```

- Can you write `fold_right` (or `fold_left`) with just `map`? How, or why not?


# An Important Optimization

Normal  
call



- When a function call is made, the return address needs to be saved to the stack so we know to where to return when the call is finished
- What if  $f$  calls  $g$  and  $g$  calls  $h$ , but calling  $h$  is the last thing  $g$  does (a *tail call*)?

# An Important Optimization



- When a function call is made, the return address needs to be saved to the stack so we know to where to return when the call is finished
- What if *f* calls *g* and *g* calls *h*, but calling *h* is the last thing *g* does (a *tail call*)?
- Then *h* can return directly to *f* instead of *g*



# Tail Recursion

---

- A recursive program is tail recursive if all recursive calls are tail calls
- Tail recursive programs may be optimized to be implemented as loops, thus removing the function call overhead for the recursive calls
- Tail recursion generally requires extra “accumulator” arguments to pass partial results
  - May require an auxiliary function



# Tail Recursion - Example

---

```
# let rec rev_aux list revlist =  
  match list with [ ] -> revlist  
  | x :: xs -> rev_aux xs (x::revlist);;  
val rev_aux : 'a list -> 'a list -> 'a list = <fun>
```

```
# let rev list = rev_aux list [ ];;  
val rev : 'a list -> 'a list = <fun>
```

- What is its running time?



# Comparison

---

- `poor_rev [1,2,3] =`
- `(poor_rev [2,3]) @ [1] =`
- `((poor_rev [3]) @ [2]) @ [1] =`
- `((poor_rev [ ]) @ [3]) @ [2]) @ [1] =`
- `(( [ ] @ [3]) @ [2]) @ [1] =`
- `([3] @ [2]) @ [1] =`
- `(3 :: ([ ] @ [2])) @ [1] =`
- `[3,2] @ [1] =`
- `3 :: ([2] @ [1]) =`
- `3 :: (2 :: ([ ] @ [1])) = [3, 2, 1]`



# Comparison

---

- $\text{rev } [1,2,3] =$
- $\text{rev\_aux } [1,2,3] [ ] =$
- $\text{rev\_aux } [2,3] [1] =$
- $\text{rev\_aux } [3] [2,1] =$
- $\text{rev\_aux } [ ] [3,2,1] = [3,2,1]$



# Folding

---

- Can replace recursion by `fold_right` in any forward primitive recursive definition
  - Primitive recursive means it only recurses on immediate subcomponents of recursive data structure
- Can replace recursion by `fold_left` in any tail primitive recursive definition